<commit_before>@echo off

set inno_compiler=

if defined INNO_HOME (
    if exist "%INNO_HOME%\ISCC.exe" (
        set "inno_compiler=%INNO_HOME%"
    )
)

if %PROCESSOR_ARCHITECTURE%==x86 (
    set RegQry=HKLM\SOFTWARE\Microsoft\Windows\CurrentVersion\Uninstall\Inno Setup 5_is1
) else (
    set RegQry=HKLM\SOFTWARE\Wow6432Node\Microsoft\Windows\CurrentVersion\Uninstall\Inno Setup 5_is1
)

if not defined inno_compiler (
    for /F "tokens=2*" %%a in ('reg query "%RegQry%" /v InstallLocation ^|findstr InstallLocation') do set inno_compiler=%%b
)

if not defined inno_compiler (
    for %%X in (ISCC.exe) do (set inno_compiler=%%~dp$PATH:X)
)

if defined inno_compiler (
    "%inno_compiler%\ISCC.exe" %1
) else (
    echo "Error, build system cannot find NSIS! Please reinstall it, add ISCC.exe to your PATH, or defined the INNO_HOME environment variable."
)<commit_msg>rename variable<commit_after>@echo off

set inno_dir=

if defined INNO_HOME (
    if exist "%INNO_HOME%\ISCC.exe" (
        set "inno_dir=%INNO_HOME%"
    )
)

if %PROCESSOR_ARCHITECTURE%==x86 (
    set RegQry=HKLM\SOFTWARE\Microsoft\Windows\CurrentVersion\Uninstall\Inno Setup 5_is1
) else (
    set RegQry=HKLM\SOFTWARE\Wow6432Node\Microsoft\Windows\CurrentVersion\Uninstall\Inno Setup 5_is1
)

if not defined inno_dir (
    for /F "tokens=2*" %%a in ('reg query "%RegQry%" /v InstallLocation ^|findstr InstallLocation') do set inno_dir=%%b
)

if not defined inno_dir (
    for %%X in (ISCC.exe) do (set inno_dir=%%~dp$PATH:X)
)

if defined inno_dir (
    "%inno_dir%\ISCC.exe" %1
) else (
    echo "Error, build system cannot find NSIS! Please reinstall it, add ISCC.exe to your PATH, or defined the INNO_HOME environment variable."
)
{'fraction_non_alphanumeric': 0.11682242990654206, 'fraction_numerical': 0.014018691588785047, 'mean_word_length': 4.222560975609756, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '5639357', 'n_tokens_mistral': 662, 'n_tokens_neox': 625, 'n_words': 163}